

# STATISTIQUES

## *pour le second cycle*

*Les calculatrices font un travail fabuleux en calculs statistiques, mais on rencontre parfois des problèmes d'adéquations.*

*En effet, de nombreuses possibilités de la calculatrice sont hors programme d'enseignement, alors que certaines exigences du second cycle ne sont pas assurées par la machine.*

*Le propos de ce dossier concerne les*  
**SERIES  
STATISTIQUES À  
VARIABLES  
CONTINUES DONT  
LES CLASSES NE  
SONT PAS DE  
MÊME AMPLITUDE.**

Dans les problèmes des statistiques à variables continues, la calculatrice donne des résultats probants dans les circonstances suivantes:

- *On connaît la liste exhaustive des variables et on introduit toutes ces variables dans la machine.*

- *Ces variables sont réparties dans des classes de même amplitude.*

Ces conditions remplies, le calcul de la médiane et des quartiles est correct, l'histogramme est significatif.

Par contre, si:

- *les variables d'une même classe sont données par le centre de la classe et l'effectif,*

- *les classes n'ont pas toutes la même amplitude* (ce qui est souvent le cas sur les bords),

alors, le calcul de la médiane et des quartiles donne le centre des classes concernées et l'histogramme est proche du «n'importe quoi».

**Les fiches de programmation, des pages suivantes, permettent de répondre à ces besoins du programme scolaire.**

De plus les problèmes de configuration de la calculatrice n'existent plus, dans la mesure où chaque programme effectue lui-même la configuration pertinente.

Je remercie ici:

- Roger Balester, Professeur Contact d'Aix-Marseille, qui m'a beaucoup aidé dans ces travaux.
- Saïd Serbouti, Professeur Contact de Lille, qui a su me prodiguer les conseils utiles.

Il y a des calculatrices (les entrées de gamme) qui ne disposent pas de certaines fonctions sophistiquées.

Aussi, lorsque ce cas se présentera dans les fiches programmes, nous donneront les deux versions.

La distinction se fera de la manière suivante:

**GAMME 1** regroupera les  
**fx-6910G**                      **fx-6910aG**  
**GRAPH 20**                      **GRAPH 25**

**GAMME 2** regroupera les  
**CFX-9930GT**                      **CFX-9960GT**  
**CFX-9940GT**                      **CFX-9990GT**  
**fx-8930GT**                      **GRAPH 30**  
**GRAPH 60**                      **GRAPH 65**  
**GRAPH 80**                      **GRAPH 100**

Les programmes écrits pour les calculatrices GAMME 1 ou GAMME 2 sont très peu différents, et cette différence met surtout en valeur un plus grand confort de lecture des résultats (permis par l'écriture localisée des informations, pour les GAMME 2).

En tout cas, **les programmes GAMME 1 fonctionnent parfaitement avec les calculatrices GAMME 2.**

(La réciproque n'est pas toujours vraie).

Maintenant, à vos calculatrices !

N.B. Ces programmes seront téléchargeables sur le site internet de CASIO France.

Les fiches programmes proposées exploitent une série statistique à variables continues d'effectif total 60.

A titre d'autre exemple, voici les résultats obtenus, avec les programmes que nous proposons, pour la distribution suivante (il s'agit du temps, en heures par semaine, passé devant la télévision, par des élèves de lycée).


Durée (h)	Effectif $n_i$
[0 ; 2[	24
[2 ; 4[	79
[4 ; 6[	131
[6 ; 8[	155
[8 ; 10[	246
[10 ; 12[	278
[12 ; 14[	231
[14 ; 16[	137
[16 ; 20[	122
[20 ; 24[	72
[24 ; 30[	25

	List 1	List 2	List 3
1	0	2	24
2	2	4	79
3	4	6	131
4	6	8	155
5	8	10	246
6	10	12	278
7	12	14	231
8	14	16	137
9	16	20	122
10	20	24	72
11	24	30	25
12			


*Les données*

List 1	List 2	List 3	List 4
0	2	24	24
2	4	79	103
4	6	131	234
6	8	155	389
8	10	246	635
10	12	278	913
12	14	231	1144
14	16	137	1281
16	20	122	1403
20	24	72	1475
24	30	25	1500

*Effectifs cumulés croissants*




*Histogramme*



*Courbe des effectifs cumulés croissants*



*Histogramme + Courbe des effectifs*




*Courbe des effectifs*

Q1 = 7.819354839  
MÉDIANE = 10.82733813  
Q3 = 13.83549784  
  
FIN DES CALCULS

*Médiane et quartiles (numérique)*

MOYENNE = 11.19066  
VARIANCE = 26.52497  
ECART TYPE = 5.150240  
  
FIN DES CALCULS

*Moyenne et écart type (numérique)*




*Médiane et quartiles (graphique)*

- Disp -  
K = 1  
 $\bar{x} - K \cdot \sigma_n = 6.040425939$   
 $\bar{x} + K \cdot \sigma_n = 16.34090739$   
EFF INT/100 = 70.2843


*Répartition des effectifs*

- Disp -  
K = 2  
 $\bar{x} - K \cdot \sigma_n = 0.8901852108$   
 $\bar{x} + K \cdot \sigma_n = 21.49114812$   
EFF INT/100 = 94.6105



*Moyenne et écart type (Graphique)*

**SANS LES PROGRAMMES**



*de la calculatrice*

1-Variable  
 $\bar{x}$  = 11.19066666  
 $\Sigma x$  = 16786  
 $\Sigma x^2$  = 227634  
 $\sigma_n$  = 5.15024072  
 $\sigma_{n-1}$  = 5.15195833  
n = 1500  
minX = 1  
Q1 =  
Med = ?  
Q3 =  
 $\bar{x} - \sigma_n = 6.04042593$   
 $\bar{x} + \sigma_n = 16.3409073$   
maxX = 27  
Mod = 11

*Ici, la médiane et les quartiles ne sont pas calculés, car des effectifs sont supérieurs à 255.*

*(effectif maximum admis sur une ligne d'une liste).*

# HISTOGRAMME



## Problème

Obtenir un histogramme d'une série statistique à variables continues, dont les classes ne sont pas de même amplitude.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6



## Principe

### Organisation:

- On range les **bornes inférieures** des classes dans la **liste n°1**,
- On range les **bornes supérieures** des classes dans la **liste n°2**,
- On range les **effectifs** des classes dans la **liste n°3**.

### Fenêtre graphique:

- Xmin** sera la borne inférieure de la première classe (List 1, ligne 1),
- Xmax** sera la borne supérieure de la dernière classe (List 2, ligne 8),
- Ymin** sera 0,
- Ymax** sera l'effectif maximum de la list 3.

### Histogramme:

On prend en considération l'**aire de chaque rectangle**.

L'amplitude **U**, d'une classe de référence, étant définie, chaque rectangle sera caractérisé par:

- sa largeur **L** : amplitude de la classe (borne supérieure - borne inférieure),
- sa hauteur **H** : effectif de la classe, proportionnellement au rapport **U/L**.





## Utilisation

- On remplit les listes (**LIST**)
- On lance le programme (**PRGM**) **HISTOG**.

	List 1	List 2	List 3
1	0	4	5
2	4	6	4
3	6	8	6
4	8	10	9
5	10	12	16
6	12	14	8
7	14	16	6
8	16	20	6
9			



	List 1	List 2	List 3
1	0	4	5
2	4	6	4
3	6	8	6
4	8	10	9
5	10	12	16
6	12	14	8
7	14	16	6
8	16	20	6
9			



# HISTOGRAMME

## Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

Nom du programme

**HISTOG**



```
"CLASSE UNITE"?→U↓
Dim List 1→K↓
'DEFINITION V-W↓
List 1[1]→Xmin↓
List 2[K]→Xmax↓
U→Xsc↓
0→Ymin↓
Max(List 3)→Ymax↓
0→Ysc↓
'HISTOGRAMME↓
For 1→I To K↓
List 2[I]-List 1[1]→L↓
U÷(List 2[I]-List 1[I])×List 3[I]→H↓
Plot List 1[I],0↓
Plot List 1[I],H↓
Line↓
Plot L,H↓
Line↓
Plot L,0↓
Line↓
Next
```



# EFFECTIFS CUMULÉS



## Problème

Obtenir la courbe des effectifs cumulés croissants d'une série statistique à variables continues, dont les classes ne sont pas de même amplitude.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6
Effectifs cumulés	5	9	15	24	40	48	54	60



## Principe

### Organisation:

- On range les **bornes inférieures** des classes dans la **liste n°1**,
- On range les **bornes supérieures** des classes dans la **liste n°2**,
- On range les **effectifs** des classes dans la **liste n°3**.
- On calcule les **effectifs cumulés** des classes dans la **liste n°4**.

### Fenêtre graphique:

- Xmin** sera la borne inférieure de la première classe (List 1, ligne 1),
- Xmax** sera la borne supérieure de la dernière classe (List 2, ligne 8),
- Ymin** sera 0,
- Ymax** sera l'effectif maximum de la list 4.

### Courbe:


On trace les segments joignant les points:

- d'abscisse **L** : borne supérieure de la classe - première borne inférieure,
- d'ordonnée **H** : effectif cumulé de la classe.



## Utilisation


- On remplit les listes 1, 2 et 3 (**LIST**) si ce n'est déjà fait.
- On lance le programme (**PRGM**) **EFFCUMC**.



	List 3	List 4
1	5	5
2	4	9
3	6	15
4	9	24
5	16	40
6	8	48
7	6	54
8	6	60
9		

SRTA SRTD

On peut contrôler le contenu de List 4 (**LIST**) pour s'assurer que les cumuls ont été faits correctement.



List 3	List 4
5	5
4	9
6	15
9	24
16	40
8	48
6	54
6	60

ELN INS

On peut contrôler le contenu de List 4 (**LIST**) pour s'assurer que les cumuls ont été faits correctement.



# EFFECTIFS CUMULÉS

## Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

Les calculatrices **Gamme 1** ne disposant pas de l'instruction **Cuml** (cumul), On remplacera utilement la ligne:


**Cuml List 3→List 4**

par les lignes:

```
List 3→List 4
For 2→I To K
List 4[I]+List 4[I-1]→List 4[I]
Next
```


qui produiront le même effet.

**Nom du programme**  
**EFFCUMC1**




```

Cls
List 3→List 4
For 2→I To K
List 4[I]+List 4[I-1]→List 4[I]
Next
Dim List 1→K
'DEFINITION V-W
List 1[1]→Xmin
List 2[K]→Xmax
U→Xscl
0→Ymin
Max(List 4)→Ymax
Max(List 4)÷10→Yscl
'POLYGONE
Plot List 1[1],0
For 1→I To K
List 2[I]-List 1[1]→L
List 4[I]→H
Plot L,H
    
```




**Nom du programme**  
**EFFCUMC**



```

Cls
Cuml List 3→List 4
Dim List 1→K
'DEFINITION V-W
List 1[1]→Xmin
List 2[K]→Xmax
U→Xscl
0→Ymin
Max(List 4)→Ymax
Max(List 4)÷10→Yscl
'POLYGONE
Plot List 1[1],0
For 1→I To K
List 2[I]-List 1[1]→L
List 4[I]→H
Plot L,H
Line
Next
    
```




# MÉDIANE QUARTILES



## Problème

Déterminer la médiane et les quartiles d'une série statistique à variables continues.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6
Effectifs cumulés	5	9	15	24	40	48	54	60



## Principe

### Organisation:

- On range les **bornes inférieures** des classes dans la **liste n°1**,
- On range les **bornes supérieures** des classes dans la **liste n°2**,
- On range les **effectifs** des classes dans la **liste n°3**.
- On calcule les **effectifs cumulés** des classes dans la **liste n°4**.

### Calcul de la médiane:

La médiane recherchée correspond à un effectif de  $y = E/2$  (E: effectif total).

On lit la **liste n°4** jusqu'à trouver un effectif  $q \geq y$ .


$y$  est situé dans l'intervalle  $[n ; q]$ .

$x$ , la médiane, est situé dans l'intervalle  $[m ; p]$ .

Les triangles (ABC) et (AJI) sont semblables. On a:  $\frac{AB}{AJ} = \frac{BC}{JI}$

Ce qui se traduit par:  $\frac{p - m}{x - m} = \frac{q - n}{y - n}$

D'où la valeur de la médiane  $x$ :  $x = m + \frac{p - m}{q - n} (y - n)$



### Calcul des quartiles:

Les quartiles recherchés  $Q_1$  et  $Q_3$  correspondent respectivement à un effectif de:

$y_1 = E/4$  et de  $y_3 = 3E/4$  (E: effectif total).

On procède ensuite comme pour le calcul de la médiane.



On remplit les listes 1, 2 et 3 (**LIST**) si ce n'est déjà fait.

On lance le programme (**PRGM**) **MEDIANE**.



**EXE**

**EXE**

**EXE**

```
Q1 =
MEDIANE = 8
          10.75
Q3 =
          13.25
FIN CALCULS
```



```
Q1      = 8
MEDIANE = 10.75
Q3      = 13.25
FIN DES CALCULS
```



# MÉDIANE QUARTILES

## Indications

Ce programme suppose que :

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

Les calculatrices **Gamme 1** ne disposant pas de l'instruction **Cuml** (cumul), On remplacera utilement la ligne :

**Cuml List 3→List 4↓**

par les lignes :

```
List 3→List 4↓
For 2→I To K↓
List 4[I]+List 4[I-1]→List 4[I]↓
Next↓
```

qui produiront le même effet.

## Nom du programme SPMEDIAN

```
'RECHERCHE↓
0→I↓
Do↓
I+1→I↓
LpWhile List 4[I]<Y↓
'CALCUL↓
List 4[I]→Q↓
List 2[I]→P↓
List 4[I-1]→N↓
List 2[I-1]→M↓
M+(P-M)(Y-N)÷(Q-N)→X↓
```

## Nom du programme MEDIANE1

```
List 3→List 4↓
For 2→I To K↓
List 4[I]+List 4[I-1]→List 4[I]↓
Next↓
Dim List 1→K↓
'TROUVER Q1↓
List 4[K]÷4→Y↓
Prog "SPMEDIAN"↓
"Q1 =" :X↓
X→C:Y→D↓
'TROUVER MEDIANE↓
List 4[K]÷2→Y↓
Prog "SPMEDIAN"↓
"MEDIANE =" :X↓
X→E:Y→F↓
'TROUVER Q3↓
List 4[K]×3÷4→Y↓
Prog "SPMEDIAN"↓
"Q3 =" :X↓
X→G:Y→W↓
"FIN CALCULS"
```

## Nom du programme MEDIANE

```
Cuml List 3→List 4↓
Dim List 1→K↓
'TROUVER Q1↓
List 4[K]÷4→Y↓
Prog "SPMEDIAN"↓
Locate 1,2,"Q1" = "↓
Locate 11,2,X↓
X→C:Y→D↓
'TROUVER MEDIANE↓
List 4[K]÷2→Y↓
Prog "SPMEDIAN"↓
Locate 1,3,"MEDIANE ="↓
Locate 11,3,X↓
X→E:Y→F↓
'TROUVER Q3↓
List 4[K]×3÷4→Y↓
Prog "SPMEDIAN"↓
Locate 1,4,"Q3" = "↓
Locate 11,4,X↓
X→G:Y→W↓
Locate 4,6,"FIN DES CALCULS"
```



# MÉDIANE

## Aspect Graphique



### Problème

Déterminer la médiane et les quartiles d'une série statistique à variables continues et donner une représentation graphique de ces informations.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6
Effectifs cumulés	5	9	15	24	40	48	54	60



### Principe

#### Organisation:

On range les **bornes inférieures** des classes dans la **liste n°1**,

On range les **bornes supérieures** des classes dans la **liste n°2**,

On range les **effectifs** des classes dans la **liste n°3**.

**On utilise le programme MEDIANE** (voir fiche MÉDIANE QUARTILES).

On lit les informations numériques.

**On utilise le programme EFFCUMC** (voir fiche EFFECTIFS CUMULÉS).

**On trace Q1 , Q2 (la médiane), Q3.**



On remplit les listes 1, 2 et 3 (**LIST**) si ce n'est déjà fait.

On lance le programme (**PRGM**) **MEDGRPH**.




EXE

```
Q1 =
MEDIANE = 8
          10.75
Q3 =
          13.25
FIN CALCULS
```

EXE

EXE


EXE



EXE

```
- Disp -
Q1 = 8
MEDIANE = 10.75
Q3 = 13.25
FIN DES CALCULS
```

EXE



# MÉDIANE

## Aspect Graphique

### Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

L'utilisation de ce programme suppose que soient présents en mémoire les programmes suivants:

**MEDIANE1** (voir fiche MÉDIANE - QUARTILES).

**SPMEDIAN** (voir fiche MÉDIANE - QUARTILES).

**EFFCUMC1** (voir fiche EFFECTIFS CUMULÉS).

L'utilisation de ces programmes suppose que soient présents en mémoire les programmes suivants:

**MEDIANE** (voir fiche MÉDIANE - QUARTILES).

**SPMEDIAN** (voir fiche MÉDIANE - QUARTILES).

**EFFCUMC** (voir fiche EFFECTIFS CUMULÉS).

### Nom du programme **SPECRIT**

Gamme  
2

```
Text 5,5,"Q1="↓
Text 5,20,C↓
Text 12,5,"Q2="↓
Text 12,20,E↓
Text 19,5,"Q3="↓
Text 19,20,G↓
```

### Nom du programme **MEDGRPH1**

Gamme  
1

```
Prog "MEDIANE1"↓
Prog "EFFCUMC1"↓
'TACER Q1↓
Plot C,0↓
Plot C,D↓
Line↓
'TACER Q2↓
Plot E,0↓
Plot E,F↓
Line↓
'TACER Q3↓
Plot G,0↓
Plot G,W↓
Line
```

### Nom du programme **MEDGRPH**

Gamme  
2

```
Prog "MEDIANE"↓
Prog "EFFCUMC"↓
'TACER Q1↓
Plot C,0↓
Plot C,D↓
Line↓
'TACER Q2↓
Plot E,0↓
Plot E,F↓
Line↓
'TACER Q3↓
Plot G,0↓
Plot G,W↓
Line↓
Prog "SPECRIT"
```



# MOYENNE ECART TYPE



## Problème

Pour une série statistique à variables continues, dont les classes ne sont pas de même amplitude, déterminer la moyenne, la variance et l'écart type.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6



## Principe

### Organisation:

- On range les **bornes inférieures** des classes dans la **liste n°1**,
- On range les **bornes supérieures** des classes dans la **liste n°2**,
- On range les **effectifs** des classes dans la **liste n°3**.
- On calcule les **centres de classes** dans la **liste n°5**.

### Calcul des résultats:

Les variables étant ramenées au centre de leur classe (en List 5) et les effectifs copiés (en List 6), on exploite la fonction de calcul statistique de la calculatrice (avec List 5 et List 6).

On obtient directement:  $\bar{X}$ ,  $\sigma^2$ ,  $\sigma$ , que l'on affiche.



## Utilisation

- On remplit les listes (**LIST**)
- On lance le programme (**PRGM**) **MOYENNE**.

List 1	List 2	List 3
1	0	4
2	4	6
3	6	8
4	8	10
5	10	12
6	12	14
7	14	16
8	16	20
9		

List 5	List 6
2	5
5	4
7	6
9	9
11	16
13	8
15	6
18	6



List 1	List 2	List 3	List 5	List 6
1	0	4	2	5
2	4	6	5	4
3	6	8	7	6
4	8	10	9	9
5	10	12	11	16
6	12	14	13	8
7	14	16	15	6
8	16	20	18	6
9				

```
MOYENNE =
10.5166667
VARIANCE =
18.14972222
ECART TYPE =
4.26024908
FIN CALCULS
```

```
MOYENNE = 10.51666
VARIANCE = 18.14972
ECART TYPE = 4.260249
FIN DES CALCULS
```



# MOYENNE ECART TYPE

## Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

### Nom du programme MOYENNE1



```
Dim List 1→K
'CENTRES CLASSES
List 2→List 5
For 1→I To K
(List 5[I]+List 1[I])÷2
→List 5[I]
Next
'RESULTATS
List 3→List 6
1-Variable List5,List6
"MOYENNE =" :  $\bar{x}$ 
"VARIANCE =" :  $x\sigma n^2$ 
"ECART TYPE =" :  $x\sigma n$ 
"FIN CALCULS"
```

### Nom du programme MOYENNE



```
Dim List 1→K
'CENTRES CLASSES
List 2→List 5
For 1→I To K
(List 5[I]+List 1[I])÷2
→List 5[I]
Next
'RESULTATS
List 3→List 6
1-Variable List5,List6
Locate 1,2,"MOYENNE ="
Locate 14,2, $\bar{x}$ 
Locate 1,3,"VARIANCE ="
Locate 14,3, $x\sigma n^2$ 
Locate 1,4,"ECART TYPE ="
Locate 14,4, $x\sigma n$ 
Locate 4,6,"FIN DES CALCULS"
```

## Indications

Accès aux instructions:

1-Variable List5,List6

mode PRGM

→ (F4) MENU (F1) STAT (F6) CALC (F1) 1VAR  
→ (F4) MENU (F1) STAT (F3) LIST (F1) List5  
→ (F4) MENU (F1) STAT (F3) LIST (F2) List6

$\bar{x}$

→ (F4) VARS (F1) STAT (F1) X (F2)  $\bar{x}$

$x\sigma n$

→ (F4) VARS (F1) STAT (F1) X (F1)  $x\sigma n$

## Indications

Accès aux instructions:

1-Variable List5,List6

mode PRGM

→ (F4) MENU (F1) STAT (F6) CALC (F1) 1VAR  
→ (F4) MENU (F1) STAT (F3) LIST (F5) List5  
→ (F4) MENU (F1) STAT (F3) LIST (F6) List6

$\bar{x}$

→ (F4) VARS (F3) STAT (F1) X (F2)  $\bar{x}$

$x\sigma n$

→ (F4) VARS (F3) STAT (F1) X (F5)  $x\sigma n$



# INTERVALLES

## $[\bar{x} - k\sigma ; \bar{x} + k\sigma]$



### Problème

Pour une série statistique à variables continues, dont les classes ne sont pas de même amplitude, déterminer le pourcentage des effectifs intérieurs à un intervalle.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6



### Principe

#### Organisation:

On range les **bornes inférieures** des classes dans la **liste n°1**,

On range les **bornes supérieures** des classes dans la **liste n°2**,

On range les **effectifs** des classes dans la **liste n°3**.

On calcule les **effectifs cumulés** des classes dans la **liste n°4**.

#### Calcul des effectifs:

• L'effectif inférieur recherché correspond à la variable  $x = \bar{x} - \sigma$ .

On lit la **liste n°2** jusqu'à trouver une variable  $p \geq x$ .


$x$  est situé dans l'intervalle  $[m ; p]$ .

$y$ , l'effectif, est situé dans l'intervalle  $[n ; q]$ .

Les triangles (ABC) et (AJI) sont semblables. On a:  $\frac{AB}{AJ} = \frac{BC}{JI}$

$$\text{Ce qui se traduit par: } \frac{p - m}{x - m} = \frac{q - n}{y - n}$$

$$\text{D'où la valeur de l'effectif } y: y = n + \frac{q - n}{p - m}(x - m)$$



On nomme **A** l'effectif trouvé.

• L'effectif supérieur recherché correspond à la variable  $x = \bar{x} + \sigma$ .

On procède comme plus haut et on nomme **B** l'effectif trouvé.

#### Calcul du pourcentage:

Le pourcentage des effectifs intérieurs à  $[\bar{x} - \sigma ; \bar{x} + \sigma]$  est donné par:  $\frac{B - A}{E} \times 100$  ( $E$  effectif total)



### Utilisation

On remplit les listes 1, 2 et 3 (**LIST**) si ce n'est déjà fait.

On lance le programme (**PRGM**) **INTERV**.

MOYENNE = 10.5166667  
VARIANCE = 18.14972222  
ECART TYPE = 4.26024908  
FIN CALCULS

K = 2  
 $\bar{x} - K\sigma_n = 1.996168507$   
 $\bar{x} + K\sigma_n = 19.03716483$   
EFF INT/100 = 93.43422768

MOYENNE = 10.51666  
VARIANCE = 18.14972  
ECART TYPE = 4.260249  
FIN DES CALCULS

K = 1  
 $\bar{x} - K\sigma_n = 6.256417587$   
 $\bar{x} + K\sigma_n = 14.77691575$   
EFF INT/100 = 67.6024

K = 1  
 $\bar{x} - K\sigma_n = 6.256417587$   
 $\bar{x} + K\sigma_n = 14.77691575$   
EFF INT/100 = 67.6024908



K = 2  
 $\bar{x} - K\sigma_n = 1.996168507$   
 $\bar{x} + K\sigma_n = 19.03716483$   
EFF INT/100 = 93.4342

# INTERVALLES

## $[\bar{X} - k\sigma ; \bar{X} + k\sigma]$

### Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

L'utilisation de ces programmes suppose que soit présent en mémoire le programme:

- **MOYENNE** (pour Gamme 2)
- **MOYENNE1** (pour Gamme 1)

Les calculatrices **Gamme 1** ne disposant pas de l'instruction **Cuml** (cumul), On remplacera utilement la ligne:


**Cuml List 3→List 4↓**

par les lignes:

```
List 3→List 4↓
For 2→I To K↓
List 4[I]+List 4[I-1]→List 4[I]↓
Next↓
```

qui produiront le même effet.

### Nom du programme **SPINTERV**



```
'RECHERCHE↓
0→I↓
Do↓
I+1→I↓
LpWhile List 2[I]<X↓
'CALCUL↓
List 4[I]→Q↓
List 2[I]→P↓
If I=1↓
Then 0→M~N↓
Else List 4[I-1]→N↓
List 2[I-1]→M↓
IfEnd↓
N+(X-M)(Q-N)÷(P-M)→Y↓
```

### Nom du programme **INTERV1**



```
List 3→List 4↓
For 2→I To K↓
List 4[I]+List 4[I-1]→List 4[I]↓
Next↓
Prog "MOYENNE1"↓
For 1→R To 2↓
"K =" : R↓
"X-Kxσ =" : X-Rxσ↓
"X+Kxσ =" : X+Rxσ↓
X-Rxσ→X↓
Prog "SPINTERV"↓
Y→A↓
X+Rxσ→X↓
Prog "SPINTERV"↓
Y→B↓
"EF INT/100 =" :
(B-A)÷n×100↓
Next
```

### Nom du programme **INTERV**



```
Cuml List 3→List 4↓
ClrText↓
Prog "MOYENNE"↓
For 1→R To 2↓
ClrText↓
Locate 1,2,"K ="↓
Locate 5,2,R↓
Locate 1,3,"X-Kxσ ="↓
Locate 10,3,X-Rxσ↓
Locate 1,4,"X+Kxσ ="↓
Locate 10,4,X+Rxσ↓
X-Rxσ→X↓
Prog "SPINTERV"↓
Y→A↓
X+Rxσ→X↓
Prog "SPINTERV"↓
Y→B↓
Locate 1,6,"EFF INT/100 ="↓
Locate 15,6,(B-A)÷n×100↓
Next
```

# COURBE DES EFFECTIFS



## Problème

Obtenir la courbe des effectifs d'une série statistique à variables continues.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6
Centres des classes	2	5	7	9	11	13	15	18




## Principe

### Organisation:

- On range les **bornes inférieures** des classes dans la **liste n°1**,
- On range les **bornes supérieures** des classes dans la **liste n°2**,
- On range les **effectifs** des classes dans la **liste n°3**.
- On calcule les **centres des classes** dans la **liste n°5**.

### Fenêtre graphique:

- Xmin** sera la borne inférieure de la première classe (List 1, ligne 1),
- Xmax** sera la borne supérieure de la dernière classe (List 2, ligne 8),
- Ymin** sera 0,
- Ymax** sera l'effectif maximum de la list 3.



### Courbe:

L'amplitude **U**, d'une classe de référence, étant définie, on trace les segments joignant les points:


- d'abscisse **L** : centre de la classe,
- d'ordonnée **H** : effectif de la classe, proportionnellement au rapport  $U/(\text{amplitude de la classe})$ .




- On remplit les listes 1, 2 et 3 (**LIST**) si ce n'est déjà fait.
- On lance le programme (**PRGM**) **EFFECT C**.



2  
EXE



2  
EXE





# COURBE DES EFFECTIFS

## Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

### Nom du programme **EFFECT C**

```
"CLASSE UNITE"?→U↓
Dim List 1→K↓
'CENTRES CLASSES↓
List 2→List 5↓
For 1→I To K↓
(List 5[I]+List 1[I])÷2→List 5[I]↓
Next↓
'DEFINITION V-W↓
List 1[1]→Xmin↓
List 2[K]→Xmax↓
U→Xscl↓
0→Ymin↓
Max(List 3)→Ymax↓
0→Yscl↓
'POLYGONE↓
Plot List 1[1],0↓
For 1→I To K↓
List 5[I]→L↓
U÷(List 2[I]-List 1[I])×List 3[I]→H↓
Plot L,H↓
Line↓
Next↓
Plot List 2[K],0↓
Line
```

## Utilisation

COMPLEMENT: On peut superposer le polygone des effectifs à l'histogramme.

1- On lance le programme (PRGM) **EFFECT C**.



On copie l'écran obtenu en **Pic1**.

→ OPTN (F1) PICT (F1) STO (F1) Pic1

2- On lance le programme (PRGM) **HISTOG**.

3- On rappelle l'écran enregistré en **Pic1**.

→ OPTN (F1) PICT (F2) RCL (F1) Pic1



# MOYENNE

## Aspect Graphique



### Problème

Déterminer la moyenne et l'écart type d'une série statistique à variables continues et donner une représentation graphique de ces informations.

Exemple numérique:

Classes	[0 ; 4[	[4 ; 6[	[6 ; 8[	[8 ; 10[	[10 ; 12[	[12 ; 14[	[14 ; 16[	[16 ; 20]
Effectifs	5	4	6	9	16	8	6	6



### Principe

#### Organisation:

On range les **bornes inférieures** des classes dans la **liste n°1**,

On range les **bornes supérieures** des classes dans la **liste n°2**,

On range les **effectifs** des classes dans la **liste n°3**.

**On utilise le programme MOYENNE** (voir fiche MOYENNE ECART TYPE).

On lit les informations numériques.

**On utilise le programme EFFECT C** (voir fiche COURBE DES EFFECTIFS).

**On calcule les pourcentages internes** (voir fiche INTERVALLES  $[\bar{x}-k\sigma ; \bar{x}+k\sigma]$  )

**On trace les verticales en  $\bar{x}-\sigma$  ,  $\bar{x}$  ,  $\bar{x}+\sigma$  , on affiche le %age .**

**On trace les verticales en  $\bar{x}-2\sigma$  ,  $\bar{x}$  ,  $\bar{x}+2\sigma$  , on affiche le %age .**



### Utilisation


On remplit les listes 1, 2 et 3 (**LIST**) si ce n'est déjà fait.

On lance le programme (**PRGM**) **MOYGRAPH**.



MOYENNE =  
10.5166667  
VARIANCE =  
18.1497222  
ECART TYPE =  
4.26024908  
FIN CALCULS


2  
EXE



EXE

- Disp -  
MOYENNE = 10.51666  
VARIANCE = 18.14972  
ECART TYPE = 4.260249  
FIN DES CALCULS

2  
EXE



# MOYENNE

## Aspect Graphique

### Indications

Ce programme suppose que:

- Les **bornes inférieures** des classes soient enregistrées en **List 1**.
- Les **bornes supérieures** des classes soient enregistrées en **List 2**.
- Les **effectifs** des classes soient enregistrés en **List 3**.

L'utilisation de ce programme suppose que soient présents en mémoire les programmes suivants:

**MOYENNE1** (voir fiche **MOYENNE-ECART TYPE**).

**SPINTERV** (voir fiche **INTERVALLES**).

**EFFECT C** (voir fiche **COURBE DES EFFECTIFS**).

L'utilisation de ces programmes suppose que soient présents en mémoire les programmes suivants:

**MOYENNE** (voir fiche **MOYENNE-ECART TYPE**).

**SPINTERV** (voir fiche **INTERVALLES**).

**EFFECT C** (voir fiche **COURBE DES EFFECTIFS**).

### Nom du programme **MOYGRPH1**

```

List 3→List 4↓
For 2→I To K↓
List 4[I]+List 4[I-1]→List 4[I]↓
Next↓
ClrText↓
Prog "MOYENNE1"↓
Prog "EFFECT C"↓
'TRACER MOYENNE↓
Plot  $\bar{x}$ ,0↓
Plot  $\bar{x}$ ,Max(List 3)↓
Line↓
For 1→R To 2↓
 $\bar{x}-R\sigma\rightarrow X$ ↓
Prog "SPINTERV"↓
Y→A↓
 $\bar{x}+R\sigma\rightarrow X$ ↓
Prog "SPINTERV"↓
Y→B↓
(B-A)÷n×100→List 6[R]↓
'TRACER B INF↓
Plot  $\bar{x}-R\sigma$ ,0↓
Plot  $\bar{x}-R\sigma$ ,A↓
Line↓
'TRACER B SUP↓
Plot  $\bar{x}+R\sigma$ ,0↓
Plot  $\bar{x}+R\sigma$ ,A↓
Line↓
Next↓
" $\bar{x}-s$ ; $\bar{x}+s$ "→"↓
List 6[1]↓
" $\bar{x}-2s$ ; $\bar{x}+2s$ "→"↓
List 6[2]↓
    
```

### Nom du programme **SP2ECRIT**

```

Text 5,5," $\bar{x}-s$ ; $\bar{x}+s$ "→"↓
Text 5,100,List 6[1]↓
Text 12,5," $\bar{x}-2s$ ; $\bar{x}+2s$ "→"↓
Text 12,100,List 6[2]↓
    
```

### Nom du programme **MOYGRPH**

```

Cuml List 3→List 4↓
ClrText↓
Prog "MOYENNE"↓
Prog "EFFECT C"↓
'TRACER MOYENNE↓
Plot  $\bar{x}$ ,0↓
Plot  $\bar{x}$ ,Max(List 3)↓
Line↓
For 1→R To 2↓
 $\bar{x}-R\sigma\rightarrow X$ ↓
Prog "SPINTERV"↓
Y→A↓
 $\bar{x}+R\sigma\rightarrow X$ ↓
Prog "SPINTERV"↓
Y→B↓
(B-A)÷n×100→List 6[R]↓
'TRACER B INF↓
Plot  $\bar{x}-R\sigma$ ,0↓
Plot  $\bar{x}-R\sigma$ ,A↓
Line↓
'TRACER B SUP↓
Plot  $\bar{x}+R\sigma$ ,0↓
Plot  $\bar{x}+R\sigma$ ,A↓
Line↓
Next↓
Prog "SP2ECRIT"↓
    
```